
Customer lifetime value spreadsheet
formula card
Every formula from the guide, for Excel and Google Sheets, on one page.

Sheet "Sales"

Column A: date. Column B: customer (name or ID, spelled the same every time). Column C: amount. Row

1 holds headers; data from row 2.

Sheet "Summary" — cells

Cell What Formula (Excel and Sheets unless noted)

B1 Start of period 1 a date, e.g. 1 Oct 2024

B2 Start of period 2 =EDATE(B1,12)

B3 End of period 2 =EDATE(B1,24)

B5 Revenue, period 2 =SUMIFS(Sales!C:C,Sales!A:A,">="&B2,Sales!A:A,"<"&B3)

B6 Sales, period 2 =COUNTIFS(Sales!A:A,">="&B2,Sales!A:A,"<"&B3)

B7
Customers,

period 2
=COUNTIF(Customers!E:E,TRUE)

B8 Average sale =B5/B6

B9

Purchases per

customer per

year

=B6/B7

B10 Retention =COUNTIFS(Customers!D:D,TRUE,Customers!E:E,TRUE)/COUNTIF(Customers!D:D,TRUE)

B11 Yearly churn =1-B10

B12 Lifespan (years) =1/B11

B13 Gross margin type it, e.g. 60%

B14
Customer

lifetime value
=B8*B9*B12*B13

Sheet "Customers" — one row per customer



Col What Formula in row 2

A Customer list
Excel 365: =UNIQUE(FILTER(Sales!B2:B5000,Sales!B2:B5000<>""))

Sheets: =UNIQUE(FILTER(Sales!B2:B,Sales!B2:B<>""))

B Lifetime revenue =SUMIFS(Sales!C:C,Sales!B:B,A2)

C Purchases =COUNTIFS(Sales!B:B,A2)

D
Bought in period

1?

=COUNTIFS(Sales!B:B,A2,Sales!A:A,">="&Summary!$B$1,Sales!A:A,"

<"&Summary!$B$2)>0

E
Bought in period

2?

=COUNTIFS(Sales!B:B,A2,Sales!A:A,">="&Summary!$B$2,Sales!A:A,"

<"&Summary!$B$3)>0

Copy B to E down as far as the customer list goes. Check once against fynl.io/customer-lifetime-value-

calculator.

The full guide: fynl.io/blog/customer-lifetime-value-excel/ Free CLV calculator: fynl.io/customer-lifetime-value-calculator


